



DECUS

PROGRAM LIBRARY

DECUS NO.	8-401
TITLE	DICE GAME AND TIC-TAC-TOE
AUTHOR	Lyle Kline
COMPANY	Inglemoor High School Bethell, Washington
DATE	December 1, 1970
SOURCE LANGUAGE	BASIC

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

21330



...

...

...

...

...

...

...

...

DICE GAME AND TIC-TAC-TOE

DECUS Program Library Write-up

DECUS NO. 8-401

DICE GAME

ABSTRACT

Dice game is a program that simulates a craps table. It is a good demonstration program and should be treated as such. It allows for one player to make fictitious bets and roll the dice. Full playing instructions are given by the program when it is run on line with the BASIC compiler.


TIC-TAC-TOE

ABSTRACT

This is an excellent demonstration or game program. It is possible to beat the computer for once.

Program Listing


```
10 RANDOMISE
20 PRINT "WELCOME TO THE INGLEMOOR HIGHSCHOOL CRAPS TABLE."
30 PRINT
40 PRINT
50 PRINT "DO YOU WANT INSTRUCTIONS?"
60 PRINT "TYPE '1' FOR YES AND '0' FOR NO";
70 INPUT Z
80 IF Z = 0 THEN 100
90 GOSUB 590
100 PRINT
110 PRINT "HOW MUCH DO YOU BET";
120 INPUT B1
130 IF B1 = 0 THEN 640
140 IF B1 = -1 THEN 471
150 IF B1 > 10000 THEN 451
160 GOSUB 480
170 IF T = 7 THEN 390
180 IF T = 11 THEN 390
190 IF T = 12 THEN 330
200 IF T = 2 THEN 310
210 IF T = 3 THEN 340
220 IF T = 4 THEN 440
230 IF T = 10 THEN 460
240 PRINT "YOUR POINT IS:"T
250 LET D = T
260 GOSUB 480
270 IF T = 7 THEN 340
280 IF T = 11 THEN 340
290 IF T = D THEN 420
300 GOTO 260
310 PRINT "SNAKE EYES.....";
320 GOTO 340
330 PRINT "BOX CARS.....";
```



```

340 LET W = W-B1
350 PRINT "CRAP!!"
360 PRINT "YOU LOSE " ABS(W);" DOLLARS."
370 GOTO 100
390 PRINT "A NATURAL!!"
400 LET W = W+B1
410 PRINT "YOU WIN " W;" DOLLARS."
411 GOTO 100
420 PRINT "YOU'VE MADE YOUR POINT!"
430 GOTO 400
440 PRINT "LITTLE JOE ";
450 GOTO 240
451 PRINT "SORRY, BUT THERE IS A HOUSE LIMIT OF $10000."
452 GOTO 100
460 PRINT "BIG DICK ";
470 GOTO 240
471 IF W < 0 THEN 475
472 IF W = 0 THEN 477
473 PRINT "YOU ARE AHEAD " W;" DOLLARS."
474 GOTO 100
475 PRINT "YOU ARE BEHIND " ABS(W);" DOLLARS."
476 GOTO 100
477 PRINT "YOU ARE BREAKING EVEN SO FAR."
478 GOTO 100
480 REMARK ROLL DICE ROUTINE.
500 LET D1 = INT(6*RND(Z1))+1
510 LET D2 = INT(6*RND(Z2))+1
520 LET T = D1+D2
530 PRINT D1;D2;
540 IF T > 9 THEN 570
550 PRINT " = ";T;
560 RETURN
570 PRINT " = ";T;
580 RETURN


```



```
590 PRINT
600 PRINT "ON THE FIRST THROW 7'S AND 11'S WIN, WHILE 2'S AND 3'S"
610 PRINT "AND 12'S LOSE. ON FOLLOWING THROWS 7'S AND 11'S LOSE,"
620 PRINT "WHILE THE ORIGINAL POINT WINS. TO QUIT TYPE A '0' FOR"
630 PRINT "YOUR BET, OR A '-1' TO FIND THE STATE OF YOUR WINS OR"
631 PRINT "LOSES."
635 RETURN
640 END
```


Program Listing


```
1 RANDOMIZE
10 PRINT "                T I C - T A C - T O E "
20 PRINT
30 PRINT "GAME BOARD:      1 2 3"
40 PRINT "                  8 9 4"
50 PRINT "                  7 6 5"
60 PRINT
70 FOR P = 1 TO 8
80 FOR I = 1 TO 3
90 READ T(P,I)
100 NEXT I
110 NEXT P
120 LET u = 0
130 FOR S = 1 TO 9
140 FOR J = 1 TO 4
141 READ U(S,J)
150 NEXT J
160 NEXT S
170 DATA 1,2,3,8,9,4,7,6,5,1,8,7,2,9,6,3,4,5,1,9,5,7,9,3
180 DATA 1,4,7,0,1,5,0,0,1,6,8,0,2,6,0,0,3,6,7,0
190 DATA 3,5,0,0,3,4,8,0,2,4,0,0,2,5,7,8
200 LET N = 0
205 FOR S = 1 TO 9
210 LET C(S) = 0
220 LET B(S) = 0
230 NEXT S
240 IF RND(Z) < .5 THEN 280
250 PRINT "YOU MOVE FIRST."
260 PRINT
270 GOTO 310
280 PRINT "THE COMPUTER MOVES FIRST."
290 PRINT
300 GOTO 470
```



```

310 PRINT "YOUR MOVE";
320 INPUT M
330 LET F = -1
340 IF B(M) <> 0 THEN 260
350 LET B(M) = F
360 FOR J = 1 TO 4
370 LET P = U(M,J)
380 IF P = 0 THEN 420
390 LET C(P) = C(P)+F
400 IF C(P) = -3 THEN 560
410 IF C(P) = 3 THEN 510
420 NEXT J
430 LET N = N+1
440 IF N = 9 THEN 580
450 IF F = 1 THEN 310
460 GOSUB 600
480 PRINT "THE COMPUTER MOVES:" M
490 LET F = 1
500 GOTO 310
510 PRINT "AND THE COMPUTER WINS!!!"
520 PRINT
530 PRINT
540 PRINT "NEW GAME."
550 GOTO 200
560 PRINT "CONGRATULATIONS, YOU JUST BEAT THE COMPUTER!!"
570 GOTO 520
580 PRINT "THE GAME IS A DRAW."
590 GOTO 520
600 FOR P = 1 TO 8
610 IF C(P) = 2 THEN 670
620 NEXT P
630 FOR P = 1 TO 8
640 IF C(P) = -2 THEN 670
650 NEXT P
660 GOTO 710
670 FOR I = 1 TO 3
680 LET M = T(P,I)

```



```
690 IF B(M) = 0 THEN 860
700 NEXT I
710 FOR S = 1 TO 9
720 LET V(S) = 0
730 IF B(S) <> 0 THEN 790
740 FOR J = 1 TO 4
750 LET P = U(S,J)
760 IF P = 0 THEN 780
770 LET V(S) = V(S)+1+ABS(C(P))
780 NEXT J
790 NEXT S
800 LET V = 0
810 FOR S = 1 TO 9
820 IF V(S) <= V THEN 850
830 LET V = V(S)
840 LET M = S
850 NEXT S
860 RETURN
870 END
```

